

NAG Library Routine Document

F08MDF (DBDSDC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08MDF (DBDSDC) computes the singular values and, optionally, the left and right singular vectors of a real n by n (upper or lower) bidiagonal matrix B .

2 Specification

```
SUBROUTINE F08MDF (UPLO, COMPQ, N, D, E, U, LDU, VT, LDVT, Q, IQ, WORK,      &
                  IWORK, INFO)
INTEGER          N, LDU, LDVT, IQ(*), IWORK(8*N), INFO
REAL (KIND=nag_wp) D(*), E(*), U(LDU,*), VT(LDVT,*), Q(*), WORK(*)
CHARACTER(1)    UPLO, COMPQ
```

The routine may be called by its LAPACK name *dbdsdc*.

3 Description

F08MDF (DBDSDC) computes the singular value decomposition (SVD) of the (upper or lower) bidiagonal matrix B as

$$B = USV^T,$$

where S is a diagonal matrix with non-negative diagonal elements $s_{ii} = s_i$, such that

$$s_1 \geq s_2 \geq \dots \geq s_n \geq 0,$$

and U and V are orthogonal matrices. The diagonal elements of S are the singular values of B and the columns of U and V are respectively the corresponding left and right singular vectors of B .

When only singular values are required the routine uses the QR algorithm, but when singular vectors are required a divide and conquer method is used. The singular values can optionally be returned in compact form, although currently no routine is available to apply U or V when stored in compact form.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: UPLO – CHARACTER(1) *Input*
On entry: indicates whether B is upper or lower bidiagonal.
 UPLO = 'U'
 B is upper bidiagonal.

- UPLO = 'L'
B is lower bidiagonal.
 Constraint: UPLO = 'U' or 'L'.
- 2: COMPQ – CHARACTER(1) *Input*
On entry: specifies whether singular vectors are to be computed.
 COMPQ = 'N'
 Compute singular values only.
 COMPQ = 'P'
 Compute singular values and compute singular vectors in compact form.
 COMPQ = 'I'
 Compute singular values and singular vectors.
 Constraint: COMPQ = 'N', 'P' or 'I'.
- 3: N – INTEGER *Input*
On entry: *n*, the order of the matrix *B*.
 Constraint: $N \geq 0$.
- 4: D(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array D must be at least $\max(1, N)$.
On entry: the *n* diagonal elements of the bidiagonal matrix *B*.
On exit: if INFO = 0, the singular values of *B*.
- 5: E(*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the dimension of the array E must be at least $\max(1, N - 1)$.
On entry: the (*n* - 1) off-diagonal elements of the bidiagonal matrix *B*.
On exit: the contents of E are destroyed.
- 6: U(LDU,*) – REAL (KIND=nag_wp) array *Output*
Note: the second dimension of the array U must be at least $\max(1, N)$ if COMPQ = 'I', and at least 1 otherwise.
On exit: if COMPQ = 'I', then if INFO = 0, U contains the left singular vectors of the bidiagonal matrix *B*.
 If COMPQ \neq 'I', U is not referenced.
- 7: LDU – INTEGER *Input*
On entry: the first dimension of the array U as declared in the (sub)program from which F08MDF (DBSDC) is called.
 Constraints:
 if COMPQ = 'I', $LDU \geq \max(1, N)$;
 otherwise $LDU \geq 1$.
- 8: VT(LDVT,*) – REAL (KIND=nag_wp) array *Output*
Note: the second dimension of the array VT must be at least $\max(1, N)$ if COMPQ = 'I', and at least 1 otherwise.
On exit: if COMPQ = 'I', then if INFO = 0, the rows of VT contain the right singular vectors of the bidiagonal matrix *B*.

If $\text{COMPQ} \neq 'I'$, VT is not referenced.

9: LDVT – INTEGER

Input

On entry: the first dimension of the array VT as declared in the (sub)program from which F08MDF (DBDSDC) is called.

Constraints:

if $\text{COMPQ} = 'I'$, $\text{LDVT} \geq \max(1, N)$;
otherwise $\text{LDVT} \geq 1$.

10: Q(*) – REAL (KIND=nag_wp) array

Output

Note: the dimension of the array Q must be at least $\max(1, N^2 + 5N, ldq)$.

On exit: if $\text{COMPQ} = 'P'$, then if $\text{INFO} = 0$, Q and IQ contain the left and right singular vectors in a compact form, requiring $O(N \log_2 N)$ space instead of $2 \times N^2$. In particular, Q contains all the real data in the first $ldq = N \times (11 + 2 \times \text{smlsiz} + 8 \times \text{int}(\log_2(N/(\text{smlsiz} + 1))))$ elements of Q, where *smlsiz* is equal to the maximum size of the subproblems at the bottom of the computation tree (usually about 25).

If $\text{COMPQ} \neq 'P'$, Q is not referenced.

11: IQ(*) – INTEGER array

Output

Note: the dimension of the array IQ must be at least $\max(1, ldq)$.

On exit: if $\text{COMPQ} = 'P'$, then if $\text{INFO} = 0$, Q and IQ contain the left and right singular vectors in a compact form, requiring $O(N \log_2 N)$ space instead of $2 \times N^2$. In particular, IQ contains all integer data in the first $ldiq = N \times (3 + 3 \times \text{int}(\log_2(N/(\text{smlsiz} + 1))))$ elements of IQ, where *smlsiz* is equal to the maximum size of the subproblems at the bottom of the computation tree (usually about 25).

If $\text{COMPQ} \neq 'P'$, IQ is not referenced.

12: WORK(*) – REAL (KIND=nag_wp) array

Workspace

Note: the dimension of the array WORK must be at least $\max(1, 6 \times N - 2)$ if $\text{COMPQ} = 'N'$, $\max(1, 6 \times N)$ if $\text{COMPQ} = 'P'$, $\max(1, 3 \times N^2 + 4 \times N)$ if $\text{COMPQ} = 'I'$, and at least 1 otherwise.

13: IWORK($8 \times N$) – INTEGER array

Workspace

14: INFO – INTEGER

Output

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\text{INFO} < 0$

If $\text{INFO} = -i$, argument *i* had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

The algorithm failed to compute a singular value. The update process of divide-and-conquer failed.

7 Accuracy

Each computed singular value of B is accurate to nearly full relative precision, no matter how tiny the singular value. The i th computed singular value, \hat{s}_i , satisfies the bound

$$|\hat{s}_i - s_i| \leq p(n)\epsilon s_i$$

where ϵ is the *machine precision* and $p(n)$ is a modest function of n .

For bounds on the computed singular values, see Section 4.9.1 of Anderson *et al.* (1999). See also F08FLF (DDISNA).

8 Parallelism and Performance

F08MDF (DBDSDC) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08MDF (DBDSDC) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

If only singular values are required, the total number of floating-point operations is approximately proportional to n^2 . When singular vectors are required the number of operations is bounded above by approximately the same number of operations as F08MEF (DBDSQR), but for large matrices F08MDF (DBDSDC) is usually much faster.

There is no complex analogue of F08MDF (DBDSDC).

10 Example

This example computes the singular value decomposition of the upper bidiagonal matrix

$$B = \begin{pmatrix} 3.62 & 1.26 & 0 & 0 \\ 0 & -2.41 & -1.53 & 0 \\ 0 & 0 & 1.92 & 1.19 \\ 0 & 0 & 0 & -1.43 \end{pmatrix}.$$

10.1 Program Text

```
Program f08mdfe
```

```
!      F08MDF Example Program Text
!
!      Mark 25 Release. NAG Copyright 2014.
!
!      .. Use Statements ..
      Use nag_library, Only: dbdsdc, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: info, ldb, ldu, ldvt, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: b(:,,:), d(:), e(:), u(:,,:), vt(:,,:), &
                                         work(:)
      Real (Kind=nag_wp)          :: q(1)
      Integer                     :: iq(1)
      Integer, Allocatable        :: iwork(:)
```

```

! .. Executable Statements ..
Write (nout,*) 'F08MDF Example Program Results'
Write (nout,*)
Flush (nout)
! Skip heading in data file
Read (nin,*)
Read (nin,*) n
ldb = n
ldu = n
ldvt = n
Allocate (b(ldb,n),d(n),e(n-1),u(ldu,n),vt(ldvt,n),work(n*(3*n+ &
4)),iwork(8*n))

! Read the bidiagonal matrix B from data file, first
! the diagonal elements, and then the off diagonal elements

Read (nin,*) d(1:n)
Read (nin,*) e(1:n-1)

! Calculate the singular values and left and right singular
! vectors of B.

! The NAG name equivalent of dbdsdc is f08mdf
Call dbdsdc('Upper','I',n,d,e,u,ldu,vt,ldvt,q,iq,work,iwork,info)

If (info==0) Then
! Print the singular values of B.

Write (nout,*) 'Singular values of B:'
Write (nout,99999) d(1:n)
Else
Write (nout,99998) '** F08MDF/DBDSDC failed with INFO = ', info
End If

99999 Format (1X,4(3X,F11.4))
99998 Format (1X,A,I10)
End Program f08mdfe

```

10.2 Program Data

F08MDF Example Program Data

```

4                               :Value of N

3.62  -2.41  1.92  -1.43  :End of diagonal elements
1.26  -1.53  1.19       :End of off-diagonal elements

```

10.3 Program Results

F08MDF Example Program Results

```

Singular values of B:
    4.0001    3.0006    1.9960    0.9998

```
